# Maintainer: Alexander F. Rødseth <xyproto@archlinux.org>
# Contributor: razer <razer@neuf.fr>

pkgname=python-serpent
pkgver=1.41
pkgrel=3
pkgdesc='Serializer for literal Python expressions'
url='https://github.com/irmen/Serpent'
arch=(any)
license=(MIT)
depends=(python)
makedepends=(git)
checkdepends=(python-attrs python-nose python-pytz python-testtools)
source=("git+$url#commit=83d06d779c05e259bf1e020aca9b850d3c6f010b") # tag: serpent-1.41
b2sums=(SKIP)

check() {
  cd Serpent
  nosetests
}

package() {
  cd Serpent
  python setup.py install --root="$pkgdir" --optimize=1
  install -Dm644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
